Potential Flow about Body in Finite Stream

K. S. Satija*
Pioneer Service and Engineering Company, Chicago, Ill.

Potential functions and stream functions for the flow of an incompressible and inviscid fluid about a prolate spheroid placed axially symmetric in a circular duct have been derived. Two methods have been considered. The first method gives the strength of an equivalent line distribution of sources. The second method directly gives the velocity distribution on the surface of the spheroid, by solving an integral equation involving Bessel functions. A computer program for solving the integral equation involved is presented. It is concluded that the velocity and pressure distributions about the body are not significantly affected due to the finiteness of the stream, unless the ratio of the diameter of the duct to that of the body is much less than 6.

Nomenclature

a = semimajor axis of the prolate spheroid
c = distance from the origin to the focus of the spheroid
h = inner radius of a circular duct
i,n = integer suffix
I_0,I_1 = modified Bessel functions
J_0,J_1 = Bessel functions
K_1 = modified Bessel function of the first order
Q_n = Legendre function of order n
r = radial distance to a point
r_0 = radial distance to the surface of the body
R = distance between a source and an arbitrary point
$u(z)$ = axial velocity of the potential flow at the surface of the body in a duct
U = axial velocity of the potential flow at the surface of the body in infinite stream
U_0 = axial velocity of the flow upstream
z = axial coordinate of the body
z_0 = axial distance between the origin and a point on the body
ϕ = velocity potential of the potential flow
ψ = stream function
ξ = distance of a point-source on the axis from the origin
μ,ζ = confocal spheroidal coordinates
θ = angle between the tangent to the body and z axis.

Introduction

ALMOST always in assimilating model tests for boundary layer and potential flow past bodies, it so happens that the extent of the width of the stream, in which the model is tested, is finite and not infinite as is mostly the case in the prototype. Such was also the situation while studying boundary layer near the tail of an elongated body of revolution, at the Iowa Institute of Hydraulic Research, Iowa City.

As the experimental body was situated at the center of a wind tunnel and was not in an infinite stream, it was considered advisable to find the effect of tunnel walls on the potential flow past the model. For simplicity, the model was considered to be a perfect spheroid (instead of the actual elongated spheroid) and the tunnel was assumed circular (instead of the actual octagonal shape). For the purpose of finding the order of magnitude of the correction to the pressure gradient of the outer potential flow due to the effect of the noninfinite stream (due to duct), the above assumptions were considered adequate.

Received July 10, 1972; revision received August 28, 1972. This work was done at the Iowa Institute of Hydraulic Research, Iowa City. This paper is a part of the author's Ph.D. thesis at the University of Iowa, Iowa City, 1971. The author is indebted to L. Landweber for his valuable advice. The investigation was partly supported by the Naval Ship Research and Development Center.

Index categories: Hydrodynamics; Submerged Vessel Systems.

*Assistant Hydraulic Engineer.

Formulation

Method I

The spheroid is assumed to be replaced by an arbitrary line distribution of sources of strength $M(\xi)$ between the focii ($\xi = -c$ and $\xi = c$) of the spheroid at the axis of the circular duct of radius "h" having a freestream of velocity U_0 (Fig. 1).

The image system in the duct is established by using a disturbance potential ϕ_1 due to the duct in the form of integrals (or series) of Bessel functions as follows.

a) Integrals of Bessel Functions

As given in Appendix A, the solution[1] for a unit source at the origin on the axis of a duct is

$$\phi = -\frac{1}{R} - \frac{2}{\pi}\int_0^\infty K_1(\kappa h)\frac{I_0(\kappa r)}{I_1(\kappa h)}\cos\kappa z\, d\kappa$$

It can be modified for the present case to

$$\phi = U_0 z - \int_{-c}^{c}\frac{M(\xi)\,d\xi}{\{(z-\xi)^2+r^2\}^{1/2}} - \frac{2}{\pi}\int_0^\infty \left[\frac{K_1(\kappa h)}{I_1(\kappa h)}\, I_0(\kappa r)\int_{-c}^{c} M(\xi)\cos(\kappa z - \kappa\xi)d\xi\right]d\kappa \quad (1)$$

Here, the first term on the right is the potential due to the freestream. The second term on the right is the total contribution from the source distribution, $M(\xi)$. The third term is the disturbance potential ϕ_1 due to the duct found as an integral of Bessel functions by using the Fourier integral technique on the nonsingular solution of the Laplace equation in cylindrical coordinates for axisymmetric flow and the boundary conditions at the duct wall, viz., $\partial\phi/\partial r = 0;\ r = h$.

I_0, K_1, I_1 are modified Bessel functions of the first order. Let ψ be the stream function. Then we have the basic relations in axisymmetric flow,

$$\partial\phi/\partial r = -(1/r)\partial\psi/\partial z;\quad \partial\phi/\partial z = (1/r)\partial\psi/\partial r \quad (2)$$



Fig. 1 Nomenclature for spheroid in circular duct.

Applying this to Eq. (1) we get

$$\psi = \frac{U_0 r^2}{2} + \int_{-c}^{c} \frac{M(\xi)\ (\xi - z)}{\{(z-\xi)^2 + r^2\}^{1/2}} d\xi + \frac{2r}{\pi} \int_0^\infty \left[\frac{K_1(\kappa h)}{I_1(\kappa h)} I_1(\kappa r) \int_{-c}^{c} M(\xi) \sin\kappa(z-\xi) d\xi \right] d\kappa \quad (3)$$

Now, applying the boundary condition that on the surface of the model,

$$r = r_0; \quad \psi = 0 \quad (4)$$

Equation (3) gives

$$0 = \frac{U_0 r_0^2}{2} + \int_{-c}^{c} \frac{M(\xi)(\xi - z_0)}{\{(\xi - z_0)^2 + r_0^2\}^{1/2}} d\xi + \frac{2r_0}{\pi} \int_0^\infty \left[\frac{K_1(\kappa h)}{I_1(\kappa h)} I_1(\kappa r_0) \int_{-c}^{c} M(\xi) \sin\kappa(z_0 - \xi) d\xi \right] d\kappa \quad (5)$$

Equation (5) is an integral equation to be solved for $M(\xi)$. Once $M(\xi)$ is known, it is put back into Eq. (1) to get, $u = \partial\phi/\partial z$, etc. Here, z_0 is the variable axial distance between the origin and a point B on the body (Fig. 1).

b) Series of Bessel Functions

Following Levine,[2] for one source (strength M) at the axis of a circular duct, we have

$$\phi = \frac{2M}{h^2} \left[z - \sum_{i=1}^{\infty} \frac{e^{-\kappa_i z}}{\kappa_i J_0^2(\kappa_i h)} J_0(\kappa_i r) \right]; \quad z > 0 \quad (6)$$

a similar expression exists for $z < 0$.

The first term in the above equation comes from the flux for large z and the second expression consists of a series of Bessel functions from the nonsingular solution of the Laplace equation for axisymmetric flow in cylindrical coordinates.

For the present case of a closed body, the total source strength is zero, and the integral of the first term must vanish. The potential due to the source distribution will then be

$$\phi = U_0 z - \frac{2}{h^2} \sum_{i=1}^{\infty} \frac{J_0(\kappa_i r)}{\kappa_i J_0^2(\kappa_i h)} \int_{-c}^{c} M(\xi) e^{-\kappa_i(z-\xi)} d\xi; \quad z > \xi > 0 \quad (7)$$

For $z < \xi$, a change of sign in the last term will occur. Using Eq. (2), we get

$$\psi = \frac{U_0 r^2}{2} + \frac{2r}{h^2} \sum_{i=1}^{\infty} \frac{J_1(\kappa_i r)}{\kappa_i J_0^2(\kappa_i h)} \int_{-c}^{c} M(\xi) e^{-\kappa_i(z-\xi)} d\xi; \quad z > \xi > 0 \quad (8)$$

Using Eq. (4), we get the following equation to be solved for $M(\xi)$:

$$0 = \frac{U_0 r_0}{2} + \frac{2}{h^2} \sum_{i=1}^{\infty} \frac{J_1(\kappa_i r_0)}{\kappa_i J_0^2(\kappa_i h)} \int_{-c}^{c} M(\xi) e^{-\kappa_i(z_0-\xi)} d\xi; \quad z > \xi > 0 \quad (9)$$

Method II (Landweber's Method)

A tangential velocity distribution on the surface of the spheroid is assumed. The said distribution implies a vortex sheet.[3] The disturbance potential ϕ_1 due to the duct was assumed in the form of integrals of Bessel functions as in Method Ia, viz.,

$$\phi_1 = -\frac{2}{\pi} \int_0^\infty \frac{K_1(\kappa h)}{I_1(\kappa h)} I_0(\kappa r) d\kappa \int_{-1}^{1} M(\xi) \cos\kappa(\xi - z) d\xi \quad (10)$$

with the same notation, but nondimensionalized by c.

The value of the axial source distribution $M(\xi)$ was assumed to be known as a first approximation from the axisymmetric flow of an infinite stream past a prolate spheroid, which has the well-known solution

$$M(\xi) = \xi / 2\dot{Q}_1(\zeta_0) \quad \{\text{for} \quad U_0 = 1\} \quad (11)$$

Here $\dot{Q}_1(\zeta_0)$ is the derivative with respect to ζ of $Q_1(\zeta)$ at $\zeta = \zeta_0$, the confocal coordinate at the spheroid surface. $Q_1(\zeta)$ is the Legendre function of the second kind and of the first order.

Then we have

$$\phi_1 = -\frac{2}{\pi \dot{Q}_1(\zeta_0)} \int_0^\infty \frac{K_1(\kappa h)}{I_1(\kappa h)} I_0(\kappa r) \sin(\kappa z)(\sin\kappa - \kappa\cos\kappa) \frac{d\kappa}{\kappa^2} \quad (12)$$

$$\left.\frac{\partial\phi_1}{\partial z}\right|_{\substack{r=0\\ z=z_0}} = -\frac{2}{\pi\dot{Q}_1(\zeta_0)} \int_0^\infty \frac{1}{h} \frac{K_1(\kappa h)}{I_1(\kappa h)} \cos(\kappa z_0)(\sin\kappa - \kappa\cos\kappa) d\kappa$$
$$= \frac{2}{\pi\dot{Q}_1(\zeta_0)} \int_0^\infty \frac{K_1(\kappa h)}{I_1(\kappa h)} (\cos\kappa - \frac{\sin\kappa}{\kappa}) \cos(\kappa z_0) d\kappa$$
$$= F_2(z_0) \quad (13)$$

F_2 is a function of z_0. The integral involved can be computed by Laguerre quadrature.[4]

Now the integral equation given by Landweber[3] for the flow of infinite stream past a body of revolution is

$$\frac{1}{2} \int_{-a}^{a} \frac{u(z)\ r_0^2(z) \sec\theta(z)}{\{(z - z_0)^2 + r_0^2(z)\}^{3/2}} dz = 1 \quad (14)$$

where a is the semimajor axis of the spheroid, and θ is the angle which the tangent to the body makes with the z axis. Hence if we include the disturbance due to the duct, the resultant integral equation is

$$1 - \frac{1}{2} \int_{-a/c}^{a/c} \frac{u(z) \sec\theta(z)\ r_0^2(z)}{\{(z - z_0)^2 + r_0^2(z)\}^{3/2}} dz + F_2(z_0) = 0 \quad (15)$$

Here z is nondimensionalized with respect to c. If it is made nondimensional with respect to a (to be able to use Gauss Quadrature), the equation becomes

$$\frac{c}{a} \int_{-1}^{1} \frac{\{u(z') \sec\theta(z')\}\ r_0^2(z')}{\{(z' - z_0')^2 + r_0^2(z')\}^{3/2}} dz' = 2[1 + F_2(z_0)] = G(z_0') \quad (16)$$

where

$$z' = (a/c) z \quad (17)$$

G is a function of z_0'. The integral in Eq. (16) can be computed by Gauss quadrature.[4] Equation (16) is the integral equation to be solved for $u(z)\ \sec\theta$ and therefore $u(z)$, because $\sec\theta$ can be easily computed for the body shape.

Numerical Solution of Integral Equation

As method II gives the velocity distribution directly, it was considered better than method I for further numerical computations. A computer program in FORTRAN IV language was written for this and is presented in Appendix B. The method used for the solution of the integral equation is given by Landweber.[5]

The velocity distribution at the surface of the prolate spheroid in an infinite (unbound) stream was calculated from the well-known solution[6]

$$U/U_0 = (c/h_1)[\zeta_0 - Q_1(\zeta_0)/\dot{Q}_1(\zeta_0)] \tag{18}$$

Alternatively, it can simply be computed from

$$U/U_0 = (1 + C_i)\cos\theta$$

where C_i is the inertia coefficient.[6] For the present study spheroid, length/diameters = 6/1, it is found that C_i = 0.045.



Fig. 2 Comparison of potential flow velocity distribution on the surface of the spheroid in duct and without duct; duct/body diameter = 6/1.

Conclusions

Figure 2 shows a comparison of the potential velocity $u(z)$ distribution on the surface of the body as computed by using Eqs. (16) and (18). Figure 2 indicates that U/U_0 values are quite close to each other in the two cases, viz, with duct and without duct, for z/a values approaching 1. For z/a values approaching zero, U/U_0 values in the two cases differ by a maximum of 2%. Therefore, in the current research, the effect of the walls of the tunnel was only 2% or less and could be considered of negligible effect to the pressure gradient of the potential flow at the surface of the body, if the ratio of the diameter of the stream to the diameter of the body is 6:1, as for the tunnel and the model that were tested. It may however be noted that relatively thicker models will affect the pressure gradient to much larger extents, as was noticed by running the same computer program for different data. Thus, the computer program presented in the appendix can be utilized for the computation of velocity distribution in potential flow past a body in a finite stream (including potential flow outside the boundary layer in viscous flow). Then, the pressure distribution in the potential flow can easily be computed by using the Bernoulli equation, as usual.

Appendix A: Potential Function for Source in a Circular Duct

For a unit source, the velocity potential function ϕ at any point, is[7]

$$\phi = -1/R \tag{A1}$$

where, R is the distance between the source and the point.

$$R = (z^2 + r^2)^{1/2} \tag{A2}$$

If the source is kept at the center of a circular duct, the duct gives a disturbance potential ϕ_1, so that the resultant potential is

$$\phi = -1/R + \phi_1 \tag{A3}$$

There is no velocity at the wall of the duct, therefore,

$$(\partial\phi/\partial r)_{r=h} = h/(z^2 + h^2)^{3/2} + (\partial\phi_1/\partial r)_{r=h} = 0 \tag{A4}$$

Writing Fourier integral,

$$\frac{h}{(z^2 + z^2)^{3/2}} = \frac{1}{\pi}\int_0^\infty \left[\int_{-\infty}^{\infty} \frac{h}{(\xi^2 + h^2)} \cos\mathcal{K}(\xi - z)\, d\xi\right] d\mathcal{K} \tag{A5}$$

ξ is as in Fig. 1. $\mathcal{K}$ is the eigenvalue.

A comparison of Eqs. (A4) and (A5) suggests that

$$\phi_1 = -\frac{1}{\pi}\int_0^\infty I_0(\mathcal{K}r)A(\mathcal{K}) \left[\int_{-\infty}^{\infty} \frac{h}{(\xi^2 + h^2)^{3/2}} \cos\mathcal{K}(\xi - z)\, d\xi\right] d\mathcal{K} \tag{A6}$$

$I_0(\mathcal{K}r)$ is modified Bessel function of $\mathcal{K}r$.

$A(\mathcal{K})$ is a coefficient determined by actually using Eqs. (A4) and (A5). In the present case, one gets

$$\phi_1 = -\frac{1}{\pi}\int_0^\infty \frac{I_0(\mathcal{K}r)}{\mathcal{K}I_1(\mathcal{K}h)} \left[\int_{-\infty}^{\infty} \frac{h\cos\mathcal{K}(\xi - z)}{(\xi^2 + h^2)^{3/2}}\, d\xi\right] d\mathcal{K} \tag{A7}$$

Now,

$$\int_{-\infty}^{\infty} \frac{h\cos\mathcal{K}(\xi - z)}{(\xi^2 + h^2)^{3/2}} = 2h\int_0^\infty \frac{\cos\mathcal{K}\xi\,\cos\mathcal{K}z}{(\xi^2 + h^2)^{3/2}}\, d\xi \tag{A8}$$

Using the well known relation[8]

$$K_1(\mathcal{K}h) = \frac{\mathcal{K}}{h}\int_0^\infty \frac{\cos\mathcal{K}\xi}{(\xi^2 + h^2)^{3/2}}\, d\xi \tag{A9}$$

Equation (A8) can be simplified. Thus Eq. (A7) gives

$$\phi_1 = -\frac{1}{\pi}\int_0^\infty \frac{I_0(\mathcal{K}r)}{I_1(\mathcal{K}h)}\, 2\mathcal{K}K_1(\mathcal{K}h)\cos\mathcal{K}z\, d\mathcal{K} \tag{A10}$$

Hence, Eq. (A3) gives

$$\phi = -\frac{1}{R} - \frac{2}{\pi}\int_0^\infty K_1(\mathcal{K}h)\frac{I_0(\mathcal{K}r)}{I_1(\mathcal{K}h)} \cos\mathcal{K}z\, d\mathcal{K} \tag{A11}$$

Appendix B: COMPUTER PROGRAM

POTENTIAL SOLUTION DUE TO PRESENCE OF DUCT WALLS

```
C     ------------------------------------------------
C        POT FLOW ABOUT SPHEROID AXISYM IN
C          CIRCULARDUCT WITH AXIAL FLOW.
C        ASSUMING VELOCITY DISTRIBUTION ON
C          SURFACE.
C        MODIFICATION OF EQN.91 AMF USED.
C        FIRST APPROXIMATION FROM EXACT
C          SOLUTION GIVING SOURCE DENSITY
C        BETWEEN FOCII APPLIED TO
C          DISTURBANCE POT. DUE TO DUCT.
C        Z    = AXIAL COORDINATE
C        WG   = WEIGHING FACTOR—GAUSS QUAD.
C        P,WL = FACTORS—LAGUERRE QUAD.
C        AC   = LENGTH/DISTANCE BET. FOCII.
C     ------------------------------------------------
      DIMENSION Z(16),YY(16),WG(16),S1(16),
    1  S2(16),A(16),
     1B(16),E(16),C(16,16),SEC(16),VEL(16),Y(16)
     2,F(16),G(16),S3(16),WL(15),P(15),GG(16,15)
      PI=3.14159265
      AC=36.0/35.0
      BC=SQRT(AC)
      Q=0.5*ALOG((BC+1.0)/(BC-1.0))-(BC/
    1  (BC**2.0-1.0))
      READ(5,1)Z,WG
    1 FORMAT(8F10.8)
      READ(5,33)P,WL
   33 FORMAT(4F20.8)
      WRITE(6,34)Z,WG
   34 FORMAT(6H DATA/(8(F12.8,2X)))
      WRITE(6,35)P,WL
   35 FORMAT(6H DATA /(4F20.9))
C     ------------------------------------------------
C     COMPUTE RADIUS Y(I) & SECANT SEC(I)
      DO 2 I=1,16
      YY(I)=(1.0-(Z(I)*Z(I)))/36.0
      Y(I)=SQRT(YY(I))
      SEC(I)=SQRT(1.0+Z(I)*Z(I)/(1296.0*YY(I)))
      Z(I)=Z(I)*BC
      YY(I)=YY(I)*BC
    2 Y(I)=Y(I)*BC
C     ------------------------------------------------
C       LAGUERRE QUADRATURE & KERNEL
C                FUNCTION
      DO 32 I=1,16
      S3(I)=0.0
      DO 3 K=1,15
      X=P(K)*BC
      CALL BESI(X,1,BI,IER)
      CALL BESK(X,1,BK,ER)
      BK=BK*EXP(X)
      BI=BI*EXP(-X)
      T=P(K)
      GG(I,K)=BK*(COS(Z(I)*T))*(COS(T)-
    1  (SIN(T)/T))/BI
    3 S3(I)=S3(I)+GG(I,K)*WL(K)*EXP(-2.0*X)
      F(I)=(4.0/(PI*Q))*S3(I)
      G(I)=2.0*(1.0+F(I))*BC
C     ------------------------------------------------
C          GAUSS QUADRATURE & FIRST
C             APPROXIMATION A(I)
      S1(I)=0.0
      DO 4 J=1,16
      D=(Z(J)-Z(I))*(Z(J)-Z(I))
      C(I,J)=WG(J)*YY(J)/(D+YY(J))**1.5
    4 S1(I)=S1(I)+C(I,J)
   32 A(I)=G(I)/S1(I)
      WRITE(6,5)A
    5 FORMAT(16H FIRST APPROX. /(F12.8))
C     ------------------------------------------------
C        NEXT APPROXIMATIONS AND GET
C     VELOCITY DISTRIBUTION ON SURFACE
      N=1
    9 DO 39 I=1,16
      S2(I)=0.0
      DO 6 M=1,16
    6 S2(I)=S2(I)+C(I,M)*A(M)
      E(I)=S2(I)/S1(I)
      B(I)=A(I)+G(I)/S1(I)-E(I)
   39 VEL(I)=B(I)/SEC(I)
      WRITE(6,7)(I,B(I),E(I),VEL(I),I=1,16)
    7 FORMAT(42H  I      B(I)      E(I)      VEL(I)
     1/(I4,3F12.8))
C     ------------------------------------------------
C          MAKE 25 APPROXIMATIONS
      DO 8 I=1,16
    8 A(I)=B(I)
      N=N+1
      IF(N.LT.25)GO TO 9
      CALL EXIT
      END
C     ------------------------------------------------
C     COMPUTES BESSEL "I" FUNCTION FOR
C        ANY "N" PROVIDED, HERE N=1.
      SUBROUTINE BESI(X,N,BI,IER)
      IER=0
      BI=1.0
      IF(N)150,15,10
   10 IF(X)160,20,20
   15 IF(X)160,17,20
   17 RETURN
   20 TOL=1.E-6
      IF(X-12.)40,40,30
   30 IF(X-FLOAT(N))40,40,110
   40 XX=X/2.
   50 TERM=1.0
      IF(N)70,70,55
   55 DO 60 I=1,N
      FI=I
      IF(ABS(TERM)-1.E-68)56,60,60
   56 IER=3
      BI=0.0
      RETURN
   60 TERM=TERM*XX/FI
   70 BI=TERM
      XX=XX*XX
      DO 90 K=1,1000
      IF(ABS(TERM)-ABS(BI*TOL))100,100,80
   80 FK=K*(N+K)
      TERM=TERM*XX/FK
   90 BI=BI+TERM
  100 RETURN
  110 FN=4*N*N
      IF(X-170.0)115,111,111
  111 IER=4
      RETURN
  115 XX=1./(8.*X)
      TERM=1.
      BI=1.
      DO 130 K=1,30
      IF(ABS(TERM)-ABS(TOL*BI))140,140,120
  120 FK=(2*K-1)**2
      TERM=TERM*XX*(FK-FN)/FLOAT(K)
  130 BI=BI+TERM
      GO TO 40
  140 PI=3.141592653
      BI=BI*EXP(X)/SQRT(2.*PI*X)
      GO TO 100
  150 IER=1
      GO TO 100
  160 IER=2
      GO TO 100
      END
```

```
C     ---------------------------------------------------
C      COMPUTE BESSEL FUNCTION K(1)
       SUBROUTINE BESK(X,N,BK,ER)
       DIMENSION T(12)
       BK=.0
       IF(N)10,11,11
    10 ER=1
       RETURN
    11 IF(X)12,12,20
    12 ER=2
       RETURN
    20 IF(X-170.0)22,22,21
    21 ER=3
       RETURN
    22 ER=0
       IF(X-1.)36,36,25
    25 A=EXP(-X)
       B=1./X
       C=SQRT(B)
       T(1)=B
       DO 26 L=2,12
    26 T(L)=T(L-1)*B
       BK=A*(1.2533141+.46999270*T(1)-.14685830*
      1   T(2)+.12804266*T(3)
      2-.17364316*T(4)+.28476181*T(5)-.45943421*
      3   T(6)+.62833807*T(7)
      4-.60322954*T(8)+.50502386*T(9)-.25813038*
      5   T(10)+.078800012*T(11)
      6-.010824177*T(12))*C
       RETURN
    36 B=X/2.
       A=.57721566+ALOG(B)
       C=B*B
       X2J=B
       FACT=1.
       HJ=1.
       GI=1./X+X2J*(.5+A-HJ)
       DO 50 J=2,8
       X2J=X2J*C
       RJ=1./FLOAT(J)
       FACT=FACT*RJ*RJ
       HJ=HJ+RJ
    50 GI=GI+X2J*FACT*(.5+(A-HJ)*FLOAT(J))
       BK=GI
       RETURN
       END
```

References

[1]Landweber, L., "Lecture Notes," *Hydrodynamics,* The Univ. of Iowa, Iowa City, Iowa, 1969.

[2]Levine, P., "Incompressible Potential Flow about Axially Symmetric Bodies in Ducts," *Journal of the Aeronautical Sciences,* Jan. 1958, pp. 33–37.

[3]Landweber, L., "Potential Flow about Bodies of Revolution and Symmetric Two-dimensional Forms," IIHR Misc. Rept., Dec. 1959, The Univ. of Iowa, Iowa City, Iowa.

[4]Abromowitz, M. and Stegun, I. A., eds., *Handbook of Mathematical Functions with Formulas, Graphs, and Mathematical Tables,* National Bureau of Standards, Applied Mathematics Series 55, June 1964.

[5]Landweber, L., "An Iteration Formula for Fredholm Integral Equations of the First Kind," *American Journal of Mathematics,* Vol. 73, No. 3, 1951.

[6]Lamb, H., *Hydrodynamics,* 6th ed., Dover, New York.

[7]Rouse, H., ed., *Advanced Mechanics of Fluids,* Wiley, New York, 1965.

[8]Basset, A., *A Treatise on Hydrodynamics,* Deighton Bell & Co., Cambridge, Mass., 1888.